
Zusammenfassung

Metadaten sind seit Version 9 ein wichtiger Bestandteil von SAS. Neben der Management

Console oder dem Metadatenbrowser im Window Environment kann man Metadaten unter 

anderem durch SAS Base Code abfragen oder verändern.

Einige Wege der Zugriffe auf Metadaten werden aufgezeigt. Es werden Beispiele vorge-

stellt, in denen Metadaten nur mittels SAS Base Code bearbeitet werden.

1 Einleitung

Metadaten oder Metainformationen sind Daten, die Informationen über Merkmale ande-

rer Daten enthalten, aber nicht diese Daten selbst.

„Bei den durch Metadaten beschriebenen Daten handelt es sich oft um größere Daten-

sammlungen wie Dokumente, Bücher, Datenbanken oder Dateien. So werden auch An-

gaben von Eigenschaften eines einzelnen Objektes (beispielsweise Personennamen) als 

dessen Metadaten bezeichnet.“    Quelle: Wikipedia

Metadaten sind das Gerüst in SAS, sie beschreiben Daten und auch die Zusammen-

hänge verschiedener Metadatenobjekte. Allerdings kann man nur diejenigen Metadaten

sehen und auslesen, für die man die Berechtigung hat. Ein vorheriges Backup ist sinn-

voll, wenn man die Metadaten nicht nur auslesen, sondern auch modifizieren möchte. 

Einen guten Einstieg in das Thema bieten [1] und [2]. In [3] erklärt Paul Homes Meta-

datenreporting mit PROC METADATA. 

Metadaten sind strukturierte Daten, die den Ort und die Struktur von Daten beschreiben. 

Beispiele sind:

Die Konfiguration verschiedener SAS Server (Workspace Server, Stored Process 

Server)

Die Konfiguration verschiedener SAS Produkte (Enterprise Guide, Enterprise 

Miner)

User, Gruppen und Rollen

Zeiger auf von SAS verwendbare Daten (Excel-, CSV-Dateien)

Spaltenattribute von SAS-Tabellen (Formate, Längen)

2 Wege, Metadaten zu verwalten

Ein erster Weg befindet sich in der SAS Management Console, mit dem Plugin Meta-

data Manager.

Abbildung 1: Metadata Manager in SAS Management Console

Um sich mit dem Metadatenserver zu verbinden, benötigt man einige Informationen, 

wie den Namen des Metadatenservers, die Logindaten, sowie den Namen des Reposi-

torys.

options metaserver='metadata-server-hostname'
metaport=8561
metauser='userid'
metapass='password'
metarepository='Foundation';

Diese Daten gibt man unter anderem beim Metadata Browser im SAS Window En-

vironment ein.

Abbildung 2: Metadata Browser

Dort sieht man auch das zentrale Element der Metadaten, den URI (Unique Resource 

Identifier). Jedes Metadatenobjekt in SAS ist somit einer eindeutigen Ziffernfolge zuge-

ordnet.

Abbildung 3: Metadata Browser mit URI

Diese URI kann man nun auch mittels PROC METADATA ansprechen. Dazu erzeugt 

man zuerst ein XML-Request.

Abbildung 4: XML mit SAS

Im folgenden Code wird eine Anfrage nach der ID und der Emailadresse eines Users 

gestellt:

data _null_;
file request;
put '<GetMetadataObjects>';
put ' <Reposid>$METAREPOSITORY</Reposid>';
put ' <Type>Person</Type>';
put ' <Objects/>';
put ' <NS>SAS</NS>';

*Flags: + OMI_GET_METADATA(256) + OMI_XMLSELECT(128) + 
OMI_TEMPLATE(4) = 388;

put ' <Flags>388</Flags>';
put ' <Options>';
put " <XMLSelect search=""Person[@Name =: 'Hinz']""/>";
put ' <Templates>';
put ' <Person Id="" Name="">';
put ' <EmailAddresses />';
put ' </Person>';
put ' <Email Address="" />';
put ' </Templates>';
put ' </Options>';
put '</GetMetadataObjects>';

run;

Nun wird PROC METADATA ausgeführt:
proc metadata in=request out=response;
run;

Auf die Anfrage nach Namen und Emailadresse bekommt man die Antwort:

<GetMetadataObjects>
<Reposid>A0000001.A5Y8I8TA</Reposid>
<Type>Person</Type>
<Objects>
<Person Id="A5Y8I8TA.AR000006" Name="Melanie Hinz">

<EmailAddresses>
<Email Id="A5Y8I8TA.BT000001" Address="melanie.hinz@mayato.com"/>
</EmailAddresses>
</Person>
</Objects>

<!-- original request deleted -->
</GetMetadataObjects>

Mit Hilfe einer XML-Map kann man diese nun nach SAS einlesen und in Tabellen und 

Berichten weiterverarbeiten.

3 SAS Data Steps

Um als Einstieg sämtliche Objektarten zu erhalten, kann man folgenden Code ausfüh-

ren. Mit metadata_getntyp bekommt man dort die n-te Objektart ausgegeben.
data _null_;

length type $64;
do until (rc<0);

i+1;
rc=metadata_getntyp(i,type);
if rc>0 then put type;

end;
run;

Das Ergebnis ist umfangreich, hier nur ein paar Beispiele:

Directory: Physischer Ordner

ExternalTable: Externe Tabelle, z.B. Excel oder CSV

Job: DI-Studio-Job

PhysicalTable: SAS-Tabelle

Transformation: Transformation in einem Job, z.B. Extract-Knoten

Tree: Ordner im DI-Studio

An dem Beispiel einer Jobliste mit Pfaden sieht man nun, dass die Objekte verschiede-

nen Typs zusammenhängen, sowie, dass wir ihre Eigenschaften auslesen können.

Man gibt den Namen des Repositorys an und initialisiert die Variablen.

OPTIONS metarepository="Foundation";
DATA Joblist;

FORMAT name $64. 
uri_job $32. 
modified $32.
uri_tree tree_name uri_par path $255. 
;

Mit einer Schleife über alle Objekte in der SAS-Bibliothek, die vom Typ Job sind, er-

hält man den URI jedes Jobs:

DO UNTIL (lrc lt 0);
n+1;
lrc = metadata_getnobj("omsobj:Job?@Name ? ''", n, uri_job);

Zu jedem Job braucht man dessen Namen und wann er zuletzt geändert wurde:

IF lrc > 0 THEN DO;
rc = metadata_getattr(uri_job, 'Name', name);
rc = metadata_getattr(uri_job,"MetadataUpdated",modified);

Jeder Job liegt in einem Ordner, ist also mit ihm assoziiert. Dieser hat natürlich eben-

falls einen Namen:

rc = metadata_getnasn(uri_job,"Trees",1,uri_tree);
rc = metadata_getattr(uri_tree,'Name',tree_name); 

Indem man jeweils den übergeordneten Ordner bestimmt, kann man den kompletten 

Pfad aufbauen.

urc = metadata_getnasn(uri_tree,"ParentTree",1,uri_par); 
path="/"||strip(tree_name);
do while (urc>0);

uri_tree=uri_par;
rc = metadata_getattr(uri_tree,'Name',tree_name);
urc = metadata_getnasn(uri_tree,"ParentTree",1,uri_par);
path="/"||strip(tree_name)||path;

end;
OUTPUT;

END; * if lrc...;
END; * do until...;
KEEP name uri_job modified path;

run;

Führt man diesen Code aus, zum Beispiel in dem User Written Code-Knoten in einem 

DI-Studio-Job, sieht das Ergebnis wie folgt aus:

Abbildung 5: DI-Studio-Job mit Jobliste

4 Weitere Anwendungsmöglichkeiten

Den umgekehrten Weg kann man ebenso gehen. Man startet mit allen Ordnern und holt 

sich die Inhalte. Das können nicht nur Jobs sein, sondern auch andere Objekte, wie Ta-

bellen, Notizen, registrierte Dokumente, usw..

Des Weiteren gibt es Security Macros von SAS, mit denen man Access Control Settings 

auswerten kann. Ein Beispiel ist, man lässt sich zu einem Ordner die Berechtigungen 

auf einem bestimmten Ordner anzeigen. Details dazu findet man auf der SAS Home-

page.

